
Chapter 42. Fonts and Text Attributes

This chapter describes the naming conventions for fonts. Tk has a font object that you can
dynamically configure and associate with widgets. This chapter also describes other text-
related attributes such as justification, anchoring, and geometry gridding.

Fonts describe how characters look on the screen. Tk widgets like buttons, labels, and
listboxes have a font attribute that determines which font they use to display their text.
The text widget has font attributes on tags that are applied to different regions of text. Tk
has a platform-independent way to name fonts (e.g., times 12 bold), plus it gracefully
handles missing fonts. You can define named font objects and then associate those with
widgets and text tags. When the font objects are reconfigured, the widgets using them
update their display automatically. You can use the resource database to define the fonts
used in your interface.

X font names (e.g., -*-times-bold-r-normal-*-12-*) were used in versions of Tk
before 8.0, and the widgets would raise errors if a font could not be found. The X names
have a pattern matching scheme that helps avoid some missing font errors. You can still
use X font names in current versions of Tk. However, the Tk font system does not do font
substitutions if you use X font names; if you use them, you must be prepared for errors.
In general, you should use the platform-independent font names.

After describing fonts, the chapter explains a few of the widget attributes that relate to
fonts. This includes justification, anchors, and geometry gridding.

Naming a Font
There are two basic ways to name a font. You can use predefined font names (e.g.,
system), or you can specify a set of font attributes with a platform-independent name:

label .foo -text "Hello" -font {times 12 bold}

In this form, the font is specified with a three element list. The first element is the font
family, the second is the size, in points, and the third is a list of style parameters. The family
determines the basic look, such as courier or helvetica.

The complete set of style parameters are normal, bold, roman, italic, underline,
and overstrike. For example, to specify both bold and italic:

label .foo -text "Hello" -font {times 12 {bold italic}}

The font size is points, which are 1/72 inch. Tk maintains a scale factor that maps from
points to pixels. The default scale is derived from the screen resolution, and you can change
it with the tk scaling command, which is described on page 669. You can specify pixel-
based sizes with negative numbers. The advantage of points over pixels is that text appears
about the same size regardless of the screen resolution. (This works better on Windows
and Macintosh than on Unix.) However, sometimes you want to control font size relative
to other widget geometry, in which case pixel-based sizes are better.

An alternate way to name font attributes uses name-value pairs. These are summarized in
Table 42-1. The format is less compact, but it is useful for changing part of a font
configuration because you do not need to specify everything. The same specification can
be made like this:

label .foo -text "Hello" -font \
     {-family times -size 12 -weight bold -slant italic}

Table 42-1. Font attributes

-family name The name can be times, courier, helvetica, and others returned by the font families
command.

-size points The font size is given in points, which are 1/72 inch.

-weight value The value is bold or normal.

-slant value The value is roman or italic.

-underline bool If bool is true, an underline is drawn.

-overstrike bool If bool is true, an overstrike line is drawn.

Tk matches a font specification with the fonts available on your system. It will use the best
possible font, but it may have to substitute some font parameters. Tk guarantees that the
Times, Courier, and Helvetica families exist. It also understands the synonyms of
Courier New for Courier, and Arial or Geneva for Helvetica.The font actual
command returns the parameters chosen to match a font specification:

font actual {times 13 bold}
-family Times -size 13 -weight bold -slant roman
     -underline 0 -overstrike 0

The Macintosh and Windows platforms have a system-defined default size. You can get
this size by specifying a size of 0 in your specification. The system font uses this:

font actual system
-family Chicago -size 0 -weight normal -slant roman
    -underline 0 -overstrike 0

Named Fonts
You can define your own names for fonts with the font create command. Creating a
named font provides a level of indirection between the font parameters and the widgets
that use the fonts. If you reconfigure a named font, the widgets using it will update their
display automatically. This makes it easy to support a user preference for font size. For
example, we can define a font name default on all platforms:

font create default -family times -size 12

The default font can be made larger at any time with font configure. Widgets using
the fonts will update automatically:

font configure default -size 14

System Fonts
The Windows and Macintosh platforms have system-defined fonts that are used by most
applications. When you query the configuration of the Tk widgets, you will see the system
font names. The parameters for the system fonts can be tuned by the user via the system
control panel. You can find out the attributes of the system font with font actual. These
are the system fonts for each platform:

• The Windows platform supports system, systemfixed, ansi, ansifixed,
device, oemfixed. The fixed suffix refers to a font where each character is the
same size.

• The Macintosh platform has system and application.
• The UNIX platform has fixed. This is the only X font name that is guaranteed to

exist. X font names are described in the next section.

Unicode Fonts
Tk does character-by-character font substitution when it displays Unicode characters. This
supports mixed display of ASCII and Kanji characters, for example. The great thing about
this is that you do not have to worry too much about choosing fonts in the simple case. The
problem with font substitution is that it can be slow. In the worst case, Tk will query every
font installed in your system to find out whether it can display a particular character. If
you know you will be displaying characters in a particular character set, you can optimize
your interface by specifying a font that matches what you expect to display.

X Font Names

Fonts can be specified with X font names on all platforms, and you must use X font names
in versions of Tk before Tk 8.0. The name fixed is an example of a short X font name.
Other short names might include 6x12, 9x15, or times12. However, these aliases are
site-dependent. In fact, all X font names are site dependent because different fonts may
be installed on different systems. The only font guaranteed to exist on the UNIX platform
is named fixed.

The more general form of an X font name has several components that describe the font
parameters. Each component is separated by a dash, and the asterisk (*) is used for
unspecified components. Short font names are system-defined aliases for these more
complete specifications. Here is an example:

-*-times-medium-r-normal-*-18-*-*-*-*-*-iso8859-1

The components of X font names are listed in Table 42-2 in the order in which they occur
in the font specification. The table gives the possible values for the components. If there
is an ellipsis (...), then there are more possibilities, too.

Table 42-2. X Font specification components

Component Possible values

foundry adobe xerox linotype misc ...

family times helvetica lucida courier symbol ...

weight bold medium demibold demi normal book light

slant i r o

swidth normal sans narrow semicondensed

Component Possible values

adstyle sans

pixels 8 10 12 14 18 24 36 48 72 144 ...

points 0 80 100 120 140 180 240 360 480 720 ...

resx 0 72 75 100

resy 0 72 75 100

space p m c

avgWidth 73 94 124 ...

registry iso8859 xerox dec adobe jisx0208.1983 ...

encoding 1 fontspecific dectech symbol dingbats

The most common attributes chosen for a font are its family, weight, slant, and size. The
weight is usually bold or medium. The slant component is a bit cryptic, but i means
italic, r means roman (i.e., normal), and o means oblique. A given font family might have
an italic version, or an oblique version, but not both. Similarly, not all weights are offered
by all font families. Size can be specified in pixels (i.e., screen pixels) or points. Points are
meant to be independent of the screen resolution. On a 75dpi font, there are about 10 points
per pixel. Note: These "points" are different than the printer points Tk uses in screen
measurements. When you use X font names, the size of the font is not affected by the Tk
scaling factor described on page 669.

It is generally a good idea to specify just a few key components and use * for the remaining
components. The X server attempts to match the font specification with its set of installed
fonts, but it fails if there is a specific component that it cannot match. If the first or last
character of the font name is an asterisk, then that can match multiple components. The
following selects a 12-pixel times font:

*times-medium-r-*-*-12*

Two useful UNIX programs that deal with X fonts are xlsfonts and xfontsel. These are part
of the standard X11 distribution. xlsfonts simply lists the available fonts that match a given
font name. It uses the same pattern matching that the server does. Because asterisk is
special to most UNIX shells, you need to quote the font name argument if you run
xlsfonts from your shell. xfontsel has a graphical user interface and displays the font that
matches a given font name.

Font Failures before Tk 8.0
Unfortunately, if a font is missing, versions of Tk before 8.0 do not attempt to substitute
another font, not even fixed. Current versions of Tk do substitutions only if you use
platform-independent font names. Otherwise, the widget creation or reconfiguration
command raises an error if the font does not exist. Example 42-1 shows one way to deal
with missing fonts, which is to create a wrapper, the FontWidget procedure, around the
Tk widget creation routines:

Example 42-1. The FontWidget procedure handles missing fonts

proc FontWidget { args } {
   # args is a Tcl command
   if {[catch $args w]} {
      # Delete the font specified in args, if any
      set ix [lsearch $args -font]
      if {$ix >= 0} {
         set args [lreplace $args $ix [expr $ix+1]]
      }
      # This font overrides the resource database
      # The "fixed" font is UNIX-specific
      set w [eval $args {-font fixed}]
   }
   return $w
}

You call FontWidget like this:

FontWidget button .foo -text Foo -font garbage

The FontWidget procedure reverts to a default font if the widget creation command fails.
It is careful to eliminate the font specified in args, if it exists. The explicit font overrides
any setting from the resource database or the Tk defaults. Of course, widget creation might
fail for some more legitimate reason, but that is allowed to happen in the backup case.
Again, the missing font problem disappears when you use platform-independent font
names, so you only need to resort to using FontWidget in early versions of Tk.

Font Metrics
The font metrics command returns measurement information for fonts. It returns
general information about all the characters in the font:

font metrics {times 10}
-ascent 9 -descent 2 -linespace 11 -fixed 0

The fixed setting is true for fonts where each character fits into the same-sized bounding
box. The linespace is the distance between the baselines of successive lines. The
ascent and descent are illustrated in Example 42-2:

Example 42-2. Font metrics

The font measure command returns the width of a string that will be displayed in a
given font. The width does not account for heavily slanted letters that overhang their
bounding box, nor does it do anything special with tabs or newlines in the string.

The font Command
Table 42-3 summarizes the font command. In the table, font is either a description of
font parameters, a logical font name, a system font name, or an X font name. The -
displayof option applies to X where you can have windows on different displays that
can support different fonts. Note that when you delete a logical font name with font
delete, the font is not really deleted if there are widgets that use that font.

Table 42-3. The font command

font actual font ?-displayof window? ?
option?

Returns the actual parameters of font.

font configure fontname ?option? ?value
option value?

Sets or queries the parameters for fontname.

font create ?fontname? ?option value ...? Defines fontname with the specified parameters.

font delete fontname ?name2 ...? Removes the definition for the named fonts.

font families ?-displayof win? Returns the list of font families supported on the display of win.

font measure font ?-displayof win? text Returns the width of text displayed in win with font.

font metrics font ?-displayof win? ?option? The option can be -ascent, -descent, -linespace, or -
fixed.

font names Returns the names of defined fonts.

Text Attributes

Layout
Table 42-4 summarizes two simple text layout attributes: justify and wrapLength. The
text widget has several more layout-related attributes, and Chapter 36 describes those
in detail. The two attributes described in this section apply to the various button widgets,
the label, entry, and message widgets. Those widgets are described in Chapters 30, 32, and
34. The justify attribute causes text to be centered, left-justified, or right-justified. The
default justification is center for all the widgets in the table, except for the entry widget,
which is left-justified by default.

The wrapLength attribute specifies how long a line of text is before it is wrapped onto
another line. It is used to create multiline buttons and labels. This attribute is specified in
screen units, however, not string length. It is probably easier to achieve the desired line
breaks by inserting newlines into the text for the button or label and specifying a
wrapLength of 0, which is the default.

Table 42-4. Layout attribute resource names

justify Text line justification. Values: left, center, or right.

Widgets: button, checkbutton, entry, label, menubutton,
message, and radiobutton.

wrapLength Maximum line length for text, in screen units. Widgets: button, checkbutton, label, menubutton,
and radiobutton.

Selection Attributes
Table 42-5 lists the selection-related attributes. The exportSelection attribute
controls if the selection is exported for cut and paste to other widgets. The colors for
selected text are set with selectForeground and selectBackground. The selection
is drawn in a raised relief, and the selectBorderWidth attribute affects the 3D
appearance. Choose a border width of zero to get a flat relief.

Table 42-5. Selection attribute resource names

exportSelection Share selection. Widgets: entry, canvas, listbox, and text.

selectForeground Foreground of selected text.

selectBackground Background of selected text.

selectBorderWidth Width of 3D raised border for selection highlight.

Gridding, Resizing, and Geometry
The text, listbox, and canvas widgets support geometry gridding. This is an alternate
interpretation of the main window geometry that is in terms of grid units, typically
characters, as opposed to pixels. The setGrid attribute is a boolean that indicates if
gridding should be turned on. The listbox and text widgets define a grid size that matches
their character size. Example 44-1 on page 658 sets up gridding for a canvas.

When a widget is gridded, its size is constrained to have a whole number of grid units
displayed. The height will be constrained to show a whole number of text lines, and the
width will be constrained to show a whole number of average width characters. This affects
interactive resizing by users, as well as the various window manger commands (wm) that
relate to geometry. When gridding is turned on, the geometry argument (e.g., 24x80) is
interpreted as grid units; otherwise, it is interpreted as pixels. The window manager
geometry commands are summarized in Table 44-1 on page 659.

The following example creates a listbox with gridded geometry enabled. Try resizing the
window in the following example with and without the -setgrid flag, and with and
without the wm minsize command, which sets the minimum size of the window. The
Scrolled_Listbox procedure is defined in Example 33-3 on page 502.

Example 42-3. A gridded, resizable listbox

wm minsize . 5 3
button .quit -text Quit -command exit
pack .quit -side top -anchor e
Scrolled_Listbox .f -width 10 -height 5 -setgrid true
pack .f -side top -fill both -expand true

A Font Selection Application

This chapter concludes with an example that lets you select fonts. It is written as a dialog
that you can add to your application. The menus are tied to elements of the font array
that are used in font configure commands. The actual settings of the font are shown
above a sampler of what the font looks like. When the user clicks the OK button, the font
configuration is returned:

Example 42-4. Font selection dialog

proc Font_Select {{top .fontsel}} {
   global font

   # Create File, Font, Size, and Format menus

   toplevel $top -class Fontsel -bd 10
   set menubar [menu $top.menubar]
   $top config -menu $menubar
   foreach x {File Font Size Format} {
      set menu [menu $menubar.[string tolower $x]]
      $menubar add cascade -menu $menu -label $x
   }
   $menubar.file add command -label Reset -command FontReset
   $menubar.file add command -label OK \
      -command {set font(ok) ok}
   $menubar.file add command -label Cancel \
      -command {set font(ok) cancel}

   # The Fonts menu lists the available Font families.

   set allfonts [font families]
   set numfonts [llength $allfonts]
   set limit 20
   if {$numfonts < $limit} {

      # Display the fonts in a single menu

      foreach family $allfonts {
         $menubar.font add radio -label $family \
            -variable font(-family) \
            -value $family \
            -command FontUpdate
      }
   } else {

      # Too many fonts. Create a set of cascaded menus to 
      # display all the font possibilities

      set c 0 ; set l 0
      foreach family $allfonts {
         if {$l == 0} {
            $menubar.font add cascade -label $family... \
               -menu $menubar.font.$c
            set m [menu $menubar.font.$c]
            incr c
         }
         $m add radio -label $family \
            -variable font(-family) \
            -value $family \

            -command FontUpdate
         set l [expr ($l +1) % $limit]
      }
   }

   # Complete the other menus

   foreach size {7 8 10 12 14 18 24 36 72} {
      $menubar.size add radio -label $size \
         -variable font(-size) \
         -value $size \
         -command FontUpdate
   }
   $menubar.size add command -label Other... \
         -command [list FontSetSize $top]
   $menubar.format add check -label Bold \
         -variable font(-weight) \
         -onvalue bold -offvalue normal \
         -command FontUpdate
   $menubar.format add check -label Italic \
         -variable font(-slant) \
         -onvalue italic -offvalue roman \
         -command FontUpdate
   $menubar.format add check -label underline \
      -variable font(-underline) \
      -command FontUpdate
   $menubar.format add check -label overstrike \
      -variable font(-overstrike) \
      -command FontUpdate

   # FontReset initializes the font array, which causes
   # the radio menu entries to get highlighted.

   FontReset

   # This label displays the current font

   label $top.font -textvar font(name) -bd 5
   # This message displays a sampler of the font.

   message $top.msg -aspect 1000 \
                -borderwidth 10 -font fontsel \
                -text "
ABCDEFGHIJKLMNOPQRSTUVWXYZ
abcdefghijklmnopqrstuvwxyz
0123456789
!@#$%^&*()_+-=[]{};:\"''~,.<>/?\\|
"

   # Lay out the dialog

   pack $top.font $top.msg -side top
   set f [frame $top.buttons]
   button $f.ok -text Ok -command {set font(ok) 1}
   button $f.cancel -text Cancel -command {set font(ok) 0}
   pack $f.ok $f.cancel -padx 10 -side left
   pack $f -side top

   # Dialog_Wait is defined in Example 39–1 on page 606

   set font(ok) cancel
   Dialog_Wait $top font(ok)
   destroy $top
   if {$font(ok) == "ok"} {
      return [array get font -*]
   } else {
      return {}
   }
}

# FontReset recreates a default font

Chapter 42. Fonts and Text Attributes Page 11 Return to Table of Contents

Chapter 42. Fonts and Text Attributes
Practical Programming in Tcl and Tk, Fourth Edition By Ken Jones, Jeffrey Hobbs, Brent
B. Welch ISBN: 0-13-038560-3 Publisher: Prentice Hall

Prepared for Ronald Fischer, Safari ID: ronald.fischer@fusshuhn.de

Print Publication Date: 2003/06/10 User number: 628024 Copyright 2007, Safari Books Online, LLC.
This PDF is exclusively for your use in accordance with the Safari Terms of Service. No part of it may be reproduced or transmitted in any form by any means without the prior
written permission for reprints and excerpts from the publisher. Redistribution or other use that violates the fair use priviledge under U.S. copyright laws (see 17 USC107) or that
otherwise violates the Safari Terms of Service is strictly prohibited.



proc FontReset {} {
   catch {font delete fontsel}
   font create fontsel
   FontSet
}

# FontSet initializes the font array with the settings
# returned by the font actual command

proc FontSet {} {
   global font

   # The name is the font configuration information
   # with a line break so it looks nicer

   set font(name) [font actual fontsel]
   regsub -- "-slant" $font(name) "\n-slant" font(name)

   # Save the actual parameters after any font substitutions

   array set font [font actual fontsel]
}

# FontSetSize adds an entry widget to the dialog so you
# can enter a specific font size.

proc FontSetSize {top} {
   set f [frame $top.size -borderwidth 10]
   pack $f -side top -fill x
   label $f.msg -text "Size:"
   entry $f.entry -textvariable font(-size)
   bind $f.entry <Return> FontUpdate
   pack $f.msg -side left
   pack $f.entry -side top -fill x
}

# FontUpdate is called when any of the font settings
# are changed, either from the menu or FontSetSize

proc FontUpdate { } {
   global font

   # The elements of font that have a leading - are
   # used directly in the font configuration command.

   eval {font configure fontsel} [array get font -*]
   FontSet
}

