
Chapter 30. Buttons and Menus

Buttons and menus are the primary way that applications expose functions to users. This
chapter describes how to create and manipulate buttons and menus.
A button widget is associated with a Tcl command that invokes an action in the application.
The checkbutton and radiobutton widgets affect an application indirectly by controlling a
Tcl variable. A menu elaborates on this concept by organizing button-like items into related
sets, including cascaded menus. The menubutton widget is a special kind of button that
displays a menu when you click on it.
Tk 8.0 provides a cross-platform menu bar facility. The menu bar is really just a menu that is
displayed horizontally along the top of your application's main window. On the Macintosh,
the menu bar appears at the top of the screen. You define the menu bar the same on all
platforms. Tk 8.0 also uses native button and menu widgets on the Windows and Macintosh
platforms. This contributes to a native look and feel for your application. In earlier versions,
Tk displayed the widgets identically on all platforms.
Associating a command to a button is usually quite simple, as illustrated by the Tk "Hello,
World!" example:

button .hello -command {puts stdout "Hello, World!"}

This chapter describes a few useful techniques for setting up the commands in more general
cases. If you use variables inside button commands, you have to understand the scoping
rules that apply. This is the first topic of the chapter. Once you get scoping figured out, then
the other aspects of buttons and menus are quite straightforward.

Button Commands and Scope Issues
Perhaps the trickiest issue with button commands has to do with variable scoping. A button
command is executed at the global scope, which is outside of any procedure. If you create a
button while inside a procedure, then the button command executes in a different scope
later. The commands used in event bindings also execute later at the global scope.
I think of this as the "now" (i.e., button definition) and "later" (i.e., button use) scope problem.
For example, you may want to use the values of some variables when you define a button
command but use the value of other variables when the button command is used. When
these two contexts are mixed, it can be confusing. The next example illustrates the problem.
The button's command involves two variables: x and val. The global variable x is needed

later, when the button's command executes. The local variable val is needed now, in order
to define the command. Example 30-1 shows this awkward mixture of scopes:

Example 30-1. A troublesome button command

proc Trouble {args} {
   set b 0
   # Display the value of x, a global variable
   label .label -textvariable x
   set f [frame .buttons -borderwidth 10]
   # Create buttons that multiply x by their value
   foreach val $args {
      button $f.$b -text $val \
         -command "set x \[expr \$x * $val\]"
      pack $f.$b -side left
      incr b
   }
   pack .label $f
}
set x 1
Trouble -1 4 7 36

The example uses a label widget to display the current value of x. The textvariable
attribute is used so that the label displays the current value of the variable, which is always
a global variable. It is not necessary to have a global command inside Trouble because
the value of x is not used there. The button's command is executed later at the global scope.
The definition of the button's command is ugly, though. The value of the loop variable
val is needed when the button is defined, but the rest of the substitutions need to be
deferred until later. The variable substitution of $x and the command substitution of expr
are suppressed by quoting with backslashes:

set x \[expr \$x * $val\]

In contrast, the following command assigns a constant expression to x each time the button
is clicked, and it depends on the current value of x, which is not defined the first time through
the loop. Clearly, this is incorrect:

button $f.$b -text $val \
    -command "set x [expr $x * $val]"

Another incorrect approach is to quote the whole command with braces. This defers too
much, preventing the value of val from being used at the correct time.

Use procedures for button commands.

The general technique for dealing with these sorts of scoping problems is to introduce Tcl
procedures for use as the button commands. Example 30-2 introduces a little procedure to
encapsulate the expression:

Example 30-2. Fixing the troublesome situation

proc LessTrouble { args } {
   set b 0
   label .label -textvariable x
   set f [frame .buttons -borderwidth 10]
   foreach val $args {
      button $f.$b -text $val \
         -command "UpdateX $val"
      pack $f.$b -side left
      incr b
   }
   pack .label $f
}
proc UpdateX { val } {
   global x
   set x [expr $x * $val]
}
set x 1
LessTrouble -1 4 7 36

It may seem just like extra work to introduce the helper procedure, UpdateX. However, it
makes the code clearer in two ways. First, you do not have to struggle with backslashes to
get the button command defined correctly. Second, the code is much clearer about the
function of the button. Its job is to update the global variable x.
You can generalize UpdateX to work on any variable by passing the name of the variable to
update. Now it becomes much like the incr command:

button $f.$b -text $val -command "Update x $val"

The definition of Update uses upvar, which is explained on page 91, to manipulate the
named variable in the global scope:

proc Update {varname val} {
   upvar #0 $varname x
   set x [expr $x * $val]
}

Double quotes are used in the button command to allow $val to be substituted. Whenever
you use quotes like this, you have to be aware of the possible values for the substitutions. If

you are not careful, the command you create may not be parsed correctly. The safest way to
generate the command is with list:

button $f.$b -text $val -command [list UpdateX $val]

Using list ensures that the command is a list of two elements, UpdateX and the value of
val. This is important because UpdateX takes only a single argument. If val contained
white space, then the resulting command would be parsed into more words than you
expected. Of course, in this case we plan to always call LessTrouble with an integer value,
which does not contain white space.
Example 30-3 provides a more straightforward application of procedures for button
commands. In this case the advantage of the procedure MaxLineLength is that it creates
a scope for the local variables used during the button action. This ensures that the local
variables do not accidentally conflict with global variables used elsewhere in the program.
There is also the standard advantage of a procedure, which is that you may find another use
for the action in another part of your program.

Example 30-3. A button associated with a Tcl procedure

proc MaxLineLength { file } {
   set max 0
   if [catch {open $file} in] {
      return $in
}
   foreach line [split [read $in] \n] {
      set len [string length $line]
      if {$len > $max} {
         set max $len
      }
   }
   return "Longest line is $max characters"
}
# Create an entry to accept the file name,
# a label to display the result
# and a button to invoke the action
. config -borderwidth 10
entry .e -width 30 -bg white -relief sunken
button .doit -text "Max Line Length" \
   -command {.label config -text [MaxLineLength [.e get]]}
label .label -text "Enter file name"
pack .e .doit .label -side top -pady 5

The example is centered around the MaxLineLength procedure. This opens a file and loops
over the lines finding the longest one. The file open is protected with catch in case the user

enters a bogus file name. In that case, the procedure returns the error message from open.
Otherwise, the procedure returns a message about the longest line in the file. The local
variables in, max, and len are hidden inside the scope of the procedure.
The user interface has three widgets: an entry for user input, the button, and a label to display
the result. These are packed into a vertical stack, and the main window is given a border.
Obviously, this simple interface can be improved in several ways. There is no Quit button,
for example.
All the action happens in the button command:

.label config -text [MaxLineLength [.e get]]

Braces are used when defining the button command so that the command substitutions all
happen when the button is clicked. The value of the entry widget is obtained with .e
get. This value is passed into MaxLineLength, and the result is configured as the text for
the label. This command is still a little complex for a button command. For example, suppose
you wanted to invoke the same command when the user pressed <Return> in the entry.
You would end up repeating this command in the entry binding. It might be better to
introduce a one-line procedure to capture this action so that it is easy to bind the action to
more than one user action. Here is how that might look:

proc Doit {} {
    .label config -text [MaxLineLength [.e get]]
}
button .doit -text "Max Line Length" -command Doit
bind .e <Return> Doit

Chapter 29 describes the bind command in detail, Chapter 32 describes the label widget,
and Chapter 35 describes the entry widget.

Buttons Associated with Tcl Variables

The checkbutton and radiobutton widgets are associated with a global Tcl variable. When
one of these buttons is clicked, a value is assigned to the Tcl variable. In addition, if the variable
is assigned a value elsewhere in the program, the appearance of the checkbutton or
radiobutton is updated to reflect the new value. A set of radiobuttons all share the same
global variable. The set represents a choice among mutually exclusive options. In contrast,
each checkbutton has its own global variable.
The ShowChoices example uses a set of radiobuttons to display a set of mutually exclusive
choices in a user interface. The ShowBooleans example uses checkbutton widgets:

Example 30-4. Radiobuttons and checkbuttons

proc ShowChoices { parent varname args } {
   set f [frame $parent.choices -borderwidth 5]
   set b 0
   foreach item $args {
      radiobutton $f.$b -variable $varname \
         -text $item -value $item
      pack $f.$b -side left
      incr b
   }
   pack $f -side top
}
proc ShowBooleans { parent args } {
   set f [frame $parent.booleans -borderwidth 5]
   set b 0
   foreach item $args {
      checkbutton $f.$b -text $item -variable $item
      pack $f.$b -side left
      incr b
   }
   pack $f -side top
}
set choice kiwi
ShowChoices {} choice apple orange peach kiwi strawberry
set Bold 1 ; set Italic 1
ShowBooleans {} Bold Italic Underline

The ShowChoices procedure takes as arguments the parent frame, the name of a variable,
and a set of possible values for that variable. If the parent frame is null, {}, then the interface
is packed into the main window. ShowChoices creates a radiobutton for each value, and it
puts the value into the text of the button. It also has to specify the value to assign to the
variable when the button is clicked because the default value associated with a radiobutton
is the empty string.
The ShowBooleans procedure is similar to ShowChoices. It takes a set of variable names
as arguments, and it creates a checkbutton for each variable. The default values for the
variable associated with a checkbutton are zero and one, which is fine for this example. If
you need particular values, you can specify them with the -onvalue and -offvalue
options.
Radiobuttons and checkbuttons can have commands associated with them, just like ordinary
buttons. The command is invoked after the associated Tcl variable has been updated.
Remember that the Tcl variable associated with the button is defined in the global scope.
For example, you could log the changes to variables as shown in the next example.

Chapter 30. Buttons and Menus Page 6 Return to Table of Contents

Chapter 30. Buttons and Menus
Practical Programming in Tcl and Tk, Fourth Edition By Ken Jones, Jeffrey Hobbs, Brent B. Welch ISBN:
0-13-038560-3 Publisher: Prentice Hall

Prepared for Ronald Fischer, Safari ID: ronald.fischer@fusshuhn.de

Print Publication Date: 2003/06/10 User number: 628024 Copyright 2007, Safari Books Online, LLC.
Reproduction, transmission and/or redistribution in any form by any means without the prior written permission from the publisher is prohibited.



Example 30-5. A command on a radiobutton or checkbutton

proc PrintByName { varname } {
   upvar #0 $varname var
   puts stdout "$varname = $var"
}
checkbutton $f.$b -text $item -variable $item \
   -command [list PrintByName $item]
radiobutton $f.$b -variable $varname \
   -text $item -value $item \
   -command [list PrintByName $varname]

Button Attributes
Table 30-1 lists the attributes for the button, checkbutton, menubutton, and radiobutton
widgets. Unless otherwise indicated, the attributes apply to all of these widget types.
Chapters 40, 41, and 42 discuss many of these attributes in more detail. Some attributes are
ignored on the Windows and Macintosh platforms because they are not supported by the
native button widgets.
The table uses the resource name for the attributes, which has capitals at internal word
boundaries. In Tcl commands, the attributes are specified with a dash and they are all
lowercase. Compare:

option add *Menubutton.activeBackground: red
.mb configure -activebackground red

The first command defines a resource database entry that covers all menubuttons and gives
them a red active background. This only affects menubuttons created after the database
entry is added. The second command changes an existing menubutton (.mb) to have a red
active background. Note the difference in capitalization of background in the two
commands. The resource database is introduced on page 372, and Chapter 31 explains how
to use the resource database in more detail.

Table 30-1. Resource names of attributes for all button widgets

activeBackground Background color when the mouse is over the button.

activeForeground Text color when the mouse is over the button.

anchor Anchor point for positioning the text.

background The normal background color.

bitmap A bitmap to display instead of text.

borderWidth Width of the border around the button.

command Tcl command to invoke when button is clicked.

compound Where the image or bitmap should be placed relative to the text: bottom, center, left, right,
top or none (default). (Tk 8.4)

cursor Cursor to display when mouse is over the widget.

default active displays as a default button. normal and disabled display as normal button. See page 809
(Tk 8.0).

direction up, down, left, right, active. Offset direction for posting menus. menubutton. (Tk 8.0).

disabledForeground Foreground (text) color when button is disabled.

font Font for the text.

foreground Foreground (text) color. (Also fg).

height Height, in lines for text, or screen units for images.

highlightBackground Focus highlight color when widget does not have focus.

highlightColor Focus highlight color when widget has focus.

highlightThickness Width of highlight border.

image Image to display instead of text or bitmap.

indicatorOn Boolean that controls if the indicator is displayed.

checkbutton, menubutton, and radiobutton.
justify Text justification: center, left, or right.

menu Menu posted when menubutton is clicked.

offRelief Alternate relief style when the widget is deselected. checkbutton and radiobutton. (Tk 8.4)

offValue Value for Tcl variable when checkbutton is not selected.

onValue Value for Tcl variable when checkbutton is selected.

overRelief Alternate relief style when mouse is over the widget. button, checkbutton, and radiobutton.
(Tk 8.4)

padX Extra space to the left and right of the button text.

padY Extra space above and below the button text.

relief flat, sunken, raised, groove, solid or ridge.

repeatDelay The number of milliseconds a button or key must be held down before it begins to auto-repeat. For
button only. (Tk 8.4)

repeatInterval The number of milliseconds between auto-repeats. For button only. (Tk 8.4)

selectColor Color for selector. checkbutton or radiobutton.

selectImage Alternate graphic image for selector: checkbutton or
radiobutton.

state normal (enabled), disabled (deactivated), or active (when the mouse pointer is over the button).

takeFocus Control focus changes from keyboard traversal.

text Text to display in the button.

textVariable Tcl variable that has the value of the text.

underline Index of text character to underline.

value Value for Tcl variable when radiobutton is selected.

variable Tcl variable associated with the button: checkbutton or
radiobutton.

width Width in characters for text, or screen units for image. As of Tk 8.4, on Windows only, a negative value
is treated as a minimum width for button widgets only.

wrapLength Maximum character length before text is wrapped, in screen units.

Button Operations
Table 30-2 summarizes the operations on button widgets. In the table, $w is a button,
checkbutton, radiobutton, or menubutton, except when noted. For the most part, these
operations are used by the script libraries that implement the bindings for buttons. The
cget and configure operations are the most commonly used by applications.

Table 30-2. Button operations

$w cget option Returns the value of the specified attribute.

$w configure ?option? ?value? ... Queries or manipulates the configuration information for the widget.

$w deselect Deselects the radiobutton or checkbutton. Set the radiobutton variable
to the null string. Set the checkbutton variable to the off value.

$w flash Redisplays the button several times in alternate colors.

$w invoke Invokes the command associated with the button.

$w select Selects the radiobutton or checkbutton, setting the associated variable
appropriately.

$w toggle Toggles the state of the checkbutton, setting the associated variable
appropriately.

Menus and Menubuttons

A menu presents a set of button-like menu entries to users. A menu entry is not a full fledged
Tk widget. Instead, you create a menu widget and then add entries to the menu as shown in
the following examples. There are several kinds of menu entries:

• Command entries are like buttons.
• Check entries are like checkbuttons.
• Radio entries are like radiobuttons.
• Separator entries are used to visually set apart entries.
• Cascade entries are used to post submenus.

• Tear-off entries are used to detach a menu from its menu button so that it becomes a
new top-level window.

A menubutton is a special kind of button that posts (i.e., displays) a menu when you press it.
If you click on a menubutton, then the menu is posted and remains posted until you click on
a menu entry to select it, or click outside the menu to dismiss it. If you press and hold the
menubutton, then the menu is unposted when you release the mouse. If you release the
mouse over the menu, it selects the menu entry that was under the mouse.
You can have a command associated with a menubutton, too. The command is invoked
before the menu is posted, which means you can compute the menu contents when the user
presses the menubutton.
Our first menu example creates a sampler of the different entry types:

Example 30-6. A menu sampler

menubutton .mb -text Sampler -menu .mb.menu
pack .mb -padx 10 -pady 10
set m [menu .mb.menu -tearoff 1]
$m add command -label Hello! -command {puts "Hello, World!"}
$m add check -label Boolean -variable foo \
   -command {puts "foo = $foo"}
$m add separator
$m add cascade -label Fruit -menu $m.sub1
set m2 [menu $m.sub1 -tearoff 0]
$m2 add radio -label apple -variable fruit -value apple
$m2 add radio -label orange -variable fruit -value orange
$m2 add radio -label kiwi -variable fruit -value kiwi

The example creates a menubutton and two menus. The main menu .mb.menu is a child of
the menubutton .mb. This relationship is necessary so that the menu displays correctly when
the menubutton is selected. Similarly, the cascaded submenu .mb.menu.sub1 is a child of
the main menu. The first menu entry is represented by the dashed line. This is a tear-off entry
that, when selected, makes a copy of the menu in a new top-level window. This is useful if
the menu operations are invoked frequently. The -tearoff 0 argument is used when
creating the submenu to eliminate its tear-off entry.

The command, radio, and check entries are similar to the corresponding button types. The
configuration options for menu entries are similar to those for buttons. The main difference
is that the text string in the menu entry is defined with the -label option, not -text. Table
30-6 gives the complete set of options for menu entries.
The cascade menu entry is associated with another menu. It is distinguished by the small
right arrow in the entry. When you select the entry, the submenu is posted. It is possible to
have several levels of cascaded menus. There is no limit to the number of levels, except that
your users will complain if you nest too many menus.

A Menu Bar
You can create a menu bar manually by packing several menubuttons into a frame. The
default bindings on menubuttons are such that you can drag your mouse over the menu bar
and the different menus will display as you drag over their menubutton.
Tk 8.0 lets you create a menu bar as a horizontal menu that is associated with a top-level
window. On Windows and UNIX the menu is displayed along the top of the window. On
Macintosh this menu replaces the main menu along the top of the screen when the window
is activated. The menu bar menu should have all cascade entries so that when you select an
entry, another menu is displayed. This is illustrated in Example 30-7. It defines variables that
store the names of the menu widgets:

set $m [menu .menubar.m$m]

This creates a variable named File, Edit, and Help that store the names of the menu
widgets. This trick is generalized on page 470 in a package that hides the menu widget names.

Example 30-7. A menu bar in Tk 8.0

menu .menubar
# attach it to the main window
. config -menu .menubar
# Create more cascade menus
foreach m {File Edit Help} {
   set $m [menu .menubar.m$m]
   .menubar add cascade -label $m -menu .menubar.m$m
}
$File add command -label Quit -command exit
# add more menu items...

System Menus
The Tk 8.0 menu bar implementation can add entries to the Windows system menu, the
Macintosh Apple menu, and the Help menu on all platforms. This works by recognizing
special names. For example, if the menu bar is .menubar, then the special names
are .menubar.system, .menubar.apple, and .menubar.help. The Help menu is right
justified on all platforms. The Apple menu is normally used by applications for their

About... entry. The entries you add to the Apple menu are added to the top of the menu.
The System menu appears in the Windows title bar and has entries such as Close and
Minimize.

Pop-Up Menus
A pop-up menu is not associated with a menubutton. Instead, it is posted in response to a
keystroke or other event in the application. The tk_popup command posts a pop-up menu:

tk_popup menu x y ?entry?

The last argument specifies the entry to activate when the menu is posted. It is an optional
parameter that defaults to 1, which avoids the tear-off entry in position zero. The menu is
posted at the specified X and Y coordinates in its parent widget.

Option Menus
An option menu represents a choice with a set of radio entries, and it displays the current
choice in the text of the menubutton. The tk_optionMenu command creates a
menubutton and a menu full of radio entries:

tk_optionMenu w varname firstValue ?value value ...?

The first argument is the pathname of the menubutton to create. The second is the variable
name. The third is the initial value for the variable, and the rest are the other choices for the
value. The menubutton displays the current choice and a small symbol, the indicator, to
indicate it is an option menu.

Multicolumn Palette Menus
Tk 8.0 adds a -columnbreak menu entry attribute that puts the entry at the top of a new
column. This is most useful when the menu consists of several images that are arranged as
a palette. Set the entry's image with the -image attribute. You can create checkbutton and
radiobutton entries that have images and no indicator by using the -hidemargin attribute.
In this case, a selected entry is indicated by drawing a solid rectangle around it.

Menu Bindings and Events

Keyboard Traversal
The default bindings for menus allow for keyboard selection of menu entries. The selection
process is started by pressing <Alt-x>, where x is the distinguishing letter for a menubutton
or a menu bar's cascade entry. The underline attribute is used to highlight the appropriate
letter. The underline value is a number that specifies a character position, and the count

starts at zero. For example, a File menu with a highlighted F is created for a menubutton
like this:

menubutton .menubar.file -text File -underline 0 \
    -menu .menubar.file.m

If the File menu is implemented as a menu bar cascade, you create the traversal highlight
like this:

menu .mbar
. configure -menu .mbar
.mbar add cascade -label File -underline 0 \
    -menu .mbar.file

When the user types <Alt-f> over the main window, the menu is posted. The case of the
highlighted letter is not important.
After a menu is posted, the arrow keys change the selected entry. The <Up> and <Down>
keys move within a menu, and the <Left> and <Right> keys move between adjacent
menus. The bindings assume that you create your menus from left to right.
If any of the menu entries have a letter highlighted with the -underline option, typing
that letter invokes that menu entry. For example, an Export entry that is invoked by typing
x can be created like this:

.menubar.file.m add command -label Export -underline 1 \
    -command File_Export

The <space> and <Return> keys invoke the menu entry that is currently selected. The
<Escape> key aborts the menu selection and removes the menu.

Menu Virtual Events
As of Tk 8.0, a menu widget generates a <<MenuSelect>> virtual event whenever the
menu's active entry changes. The event is fired after the menu selection has changed, so the
binding action can access the new selection. The easiest way to be aware of changes to the
menu selection is to bind to this virtual event, as shown in Example 30-8. Notification like
this is useful for features such as context-sensitive help.

Example 30-8. Using the <<MenuSelect>> virtual event

proc MenuChanged {w} {
   puts "Menu $w selection: [$w entrycget active -label]"
}
bind .mbar.file <<MenuSelect>> {MenuChanged %W}

Manipulating Menus and Menu Entries

There are a number of operations that apply to menu entries. We have already introduced
the add operation. The entryconfigure operation is similar to the configure operation
for widgets. It accepts the same attribute-value pairs used when the menu entry was added.
The delete operation removes a range of menu entries. The rest of the operations are used
by the library scripts that implement the standard bindings for menus.

A menu entry is referred to by an index. The index can be numerical,
counting from zero, or symbolic. Table 30-3 summarizes the index formats.
One of the most useful indices is a pattern that matches the label in the
menu entry. The pattern matching is done with the rules of string
match. Using a pattern eliminates the need to keep track of the numerical
indices.

Table 30-3. Menu entry index keywords

index A numerical index counting from zero.

active The activated entry, either because it is under the mouse or has been activated by keyboard traversal.

end The last menu entry.

last The same as end.

none No entry at all.

@ycoord The entry under the given Y coordinate. Use @%y in bindings.

pattern A string match pattern to match the label of a menu entry.

Table 30-4 summarizes the complete set of menu operations. In the table, $w is a menu
widget.

Table 30-4. Menu operations

$w activate index Highlights the specified entry.

$w add type ?option value?... Adds a new menu entry of the specified type with the given values
for various attributes.

$w cget option Returns the value for the configuration option.

$w clone Makes a linked copy of the menu. This is used to implement tear-
offs and menu bars.

$w configure ?option? ?value?... Returns the configuration information for the menu.

$w delete i1 ?i2? Deletes the menu entries from index i1 to i2.

$w entrycget index option Returns the value of option for the specified entry.

$w entryconfigure index ?option? ?value? ... Queries or modifies the configuration information for the specified
menu entry.

$w index index Returns the numerical value of index.

$w insert type index?option value? ... Like add, but inserts the new entry after the specified index.

$w invoke index Invokes the command associated with the entry.

$w post x y Displays the menu at the specified coordinates.

$w postcascade index Displays the cascade menu from entry index.

$w type index Returns the type of the entry at index.

$w unpost Unmaps the menu.

$w yposition index Returns the Y coordinate of the top of the entry.

Menu Attributes
A menu has a few global attributes, and then each menu entry has many button-like
attributes that describe its appearance and behavior. Table 30-5 specifies the attributes that
apply globally to the menu, unless overridden by a per-entry attribute. The table uses the X
resource names, which may have a capital at interior word boundaries. In Tcl commands, use
all lowercase and a leading dash.

Table 30-5. Menu attribute resource names

activeBackground Background color when the mouse is over a menu entry.

activeBorderWidth Width of the raised border around active entries.

activeForeground Text color when the mouse is over a menu entry.

background The normal background color for menu entries.

borderWidth Width of the border around the menu (except on systems where native menus are used, such as
Windows).

cursor Cursor to display when mouse is over the menu.

disabledForeground Foreground (text) color when menu entries are disabled.

font Default font for the text.

foreground Foreground color. (Also fg).

postCommand Tcl command to run just before the menu is posted.

relief The relief style of the menu (except on systems where native menus are used such, as Windows).

selectColor Color for selector in check and radio type entries.

takeFocus Control focus changes from keyboard traversal.

tearOff True if menu should contain a tear-off entry.

tearOffCommand Command to execute when menu is torn off. Two arguments are added: the original menu and the new
tear-off.

title Title for the window created when the menu is torn off. If this is an empty string (default), the title is the
text of the menubutton or cascade item from which this menu was torn off. (Tk 8.0)

type (Read-only) normal, menubar, or tearoff. (Tk 8.0).

Table 30-6 describes the attributes for menu entries, as you would use them in a Tcl command
(i.e., all lowercase with a leading dash.) The attributes for menu entries are not supported
directly by the resource database. However, Example 31-6 on page 481 describes how you
can use the resource database for menu entries.

Table 30-6. Attributes for menu entries

-activebackground Background color when the mouse is over the entry.

-activeforeground Foreground (text) color with mouse is over the entry.

-accelerator Text to display as a reminder about keystroke binding.

-background The normal background color.

-bitmap A bitmap to display instead of text.

-columnbreak Puts the entry at the start of a new column. (Tk 8.0).

-command Tcl command to invoke when entry is invoked.

-compound Where the image or bitmap should be placed relative to the text: bottom, center, left, right, top
or none (default). (Tk 8.4)

-font Default font for the text.

-foreground Foreground color. (Also fg).

-hidemargin Suppresses the margin reserved for button indicators. (Tk 8.0).

-image Image to display instead of text or bitmap.

-indicatoron Boolean that controls if the indicator is displayed: check and radio entries.

-label Text to display in the menu entry.

-menu Menu posted when cascade entry is invoked.

-offvalue Variable value when check entry is not selected.

-onvalue Value for Tcl variable when check entry is selected.

-selectcolor Color for selector: check and radio entries.

-selectimage Alternate image to use when entry is selected: check and radio entries.

-state The state: normal, active, or disabled
-underline Index of text character to underline.

-value Value for Tcl variable when radiobutton entry is selected.

-variable Tcl variable associated with the check or radio entry.

A Menu by Name Package

If your application supports extensible or user-defined menus, it can be tedious to expose
all the details of the Tk menus. The examples in this section create a little package that lets
users refer to menus and entries by name. In addition, the package keeps keystroke
accelerators for menus consistent with bindings.
The Menu_Setup procedure initializes the package. It creates a frame to hold the set of menu
buttons, and it initializes some state variables: the frame for the menubuttons and a counter
used to generate widget pathnames. All the global state for the package is kept in the array
called menu.
The Menu procedure creates a menubutton and a menu. It records the association between
the text label of the menubutton and the menu that was created for it. This mapping is used
throughout the rest of the package so that the client of the package can refer to the menu
by its label (e.g., File) as opposed to the internal Tk pathname,
(e.g., .top.menubar.file.menu).

Example 30-9. A simple menu by name package

proc Menu_Setup { menubar } {
   global menu
   frame $menubar
   pack $menubar -side top -fill x
   set menu(menubar) $menubar
   set menu(uid) 0
}
proc Menu { label } {
   global menu
   if [info exists menu(menu,$label)] {
      error "Menu $label already defined"
   }
   # Create the menubutton and its menu
   set name $menu(menubar).mb$menu(uid)
   set menuName $name.menu
   incr menu(uid)
   set mb [menubutton $name -text $label -menu $menuName]
   pack $mb -side left
   menu $menuName -tearoff 1
   # Remember the name to menu mapping
   set menu(menu,$label) $menuName
}

These procedures are repeated in Example 30-10, except that they use the Tk 8.0 menu bar
mechanism. The rest of the procedures in the package are the same with either version of
menu bars.

Example 30-10. Using the Tk 8.0 menu bar facility

proc Menu_Setup { menubar } {
   global menu
   menu $menubar
   # Associated menu with its main window
   set top [winfo parent $menubar]
   $top config -menu $menubar
   set menu(menubar) $menubar
   set menu(uid) 0
}
proc Menu { label } {
   global menu
   if [info exists menu(menu,$label)] {
      error "Menu $label already defined"
   }
   # Create the cascade menu
   set menuName $menu(menubar).mb$menu(uid)
   incr menu(uid)
   menu $menuName -tearoff 1
   $menu(menubar) add cascade -label $label -menu $menuName
   # Remember the name to menu mapping
   set menu(menu,$label) $menuName
}

Once the menu is set up, the menu array is used to map from a menu name, like File, to the
Tk widget name such as .menubar.mb3. Even though this can be done with a couple of
lines of Tcl code, the mapping is put inside the MenuGet procedure to hide the
implementation. MenuGet uses return -code error if the menu name is unknown,
which changes the error reporting slightly as shown in Example 6-19 on page 86. If the user
specifies a bogus menu name, the undefined variable error is caught and a more informative
error is raised instead. MenuGet is private to the package, so it does not have an underscore
in its name.

Example 30-11. MenuGet maps from name to menu

proc MenuGet {menuName} {
   global menu
   if [catch {set menu(menu,$menuName)} m] {
      return -code error "No such menu: $menuName"
   }
   return $m
}

The procedures Menu_Command, Menu_Check, Menu_Radio, and Menu_Separator are
simple wrappers around the basic menu commands. They use MenuGet to map from the
menu label to the Tk widget name.

Example 30-12. Adding menu entries

proc Menu_Command { menuName label command } {
   set m [MenuGet $menuName]
   $m add command -label $label -command $command
}

proc Menu_Check { menuName label var { command {} } }
{
   set m [MenuGet $menuName]
   $m add check -label $label -command $command \
      -variable $var
}

proc Menu_Radio { menuName label var {val {}} {command {}} } {
   set m [MenuGet $menuName]
   if {[string length $val] == 0} {
      set val $label
   }
   $m add radio -label $label -command $command \
      -value $val -variable $var
}

proc Menu_Separator { menuName } {
   [MenuGet $menuName] add separator
}

Creating a cascaded menu also requires saving the mapping between the label in the cascade
entry and the Tk pathname for the submenu. This package imposes a restriction that different
menus, including submenus, cannot have the same label.

Example 30-13. A wrapper for cascade entries

proc Menu_Cascade { menuName label } {
   global menu
   set m [MenuGet $menuName]
   if [info exists menu(menu,$label)] {
      error "Menu $label already defined"
   }
   set sub $m.sub$menu(uid)
   incr menu(uid)
   menu $sub -tearoff 0
   $m add cascade -label $label -menu $sub
   set menu(menu,$label) $sub
}

Creating the sampler menu with this package looks like this:

Example 30-14. Using the menu by name package

Menu_Setup .menubar
Menu Sampler
Menu_Command Sampler Hello! {puts "Hello, World!"}
Menu_Check Sampler Boolean foo {puts "foo = $foo"}
Menu_Separator Sampler
Menu_Cascade Sampler Fruit
Menu_Radio Fruit apple fruit
Menu_Radio Fruit orange fruit
Menu_Radio Fruit kiwi fruit

Menu Accelerators
The final touch on the menu package is to support accelerators in a
consistent way. A menu entry can display another column of information
that is assumed to be a keystroke identifier to remind users of a binding
that also invokes the menu entry. However, there is no guarantee that this
string is correct, or that if the user changes the binding that the menu will
be updated. Example 30-15 shows the Menu_Bind procedure that takes
care of this.

Example 30-15. Keeping the accelerator display up to date

proc Menu_Bind { what sequence accText menuName label } {
   variable menu
   set m [MenuGet $menuName]
   if {[catch {$m index $label} index]} {
      error "$label not in menu $menuName"
   }
   bind $what $sequence [list MenuInvoke $m $index]
   $m entryconfigure $index -accelerator $accText
}
proc MenuInvoke {m index} {
   set state [$m entrycget $index -state]
   if {[string equal $state normal]} {
      $m invoke $index
   }
}

The Menu_Bind command uses the index operation to find out what menu entry has the
given label. It sets up a binding for the key sequence that will invoke the menu operation,
and it updates the display of the accelerator using the entryconfigure operation. This
approach has the advantage of keeping the keystroke command consistent with the menu
command, as well as updating the display.
The MenuInvoke procedure is used for the binding. We could use entrycget to fetch the
command, and then bind directly to that. However, that wouldn't honor the state of the menu
entry, which could be temporarily disabled. In addition, the invoke operation on the menu
handles any special cases such as updating radiobutton variables associated with the entry.

To try Menu_Bind, add an empty frame to the sampler example, and bind a keystroke to it
and one of the menu commands, like this:

frame .body -width 100 -height 50
pack .body ; focus .body
Menu_Bind .body <Control-q> Ctrl-Q Sampler Hello!

