
Chapter 38. Selections and the Clipboard>

Cut and paste allows information exchange between applications, and it is built upon a
general purpose selection mechanism. The CLIPBOARD selection is used to implement cut
and paste on all platforms. X Windows applications may also use the PRIMARY selection.
This chapter describes the selection and clipboard commands.
Copy and paste is a basic way to transfer data between just about any two applications. In
Tk, copy and paste is based on a general selection mechanism where the selection has a
name, type, format, and value. For the most part you can ignore these details because they
are handled by the Tk widgets. However, you can also control the selection explicitly. This
chapter describes the selection model and the selection and clipboard commands.
The last section of this chapter presents an example that implements copy and paste of
graphical objects in a canvas.

The Selection Model

The Windows and Macintosh selection model is simpler than the selection model used in X
windows. In the Macintosh and Windows there is one selection, although that selection may
store different types of data like text or images. Users copy data from an application into a
clipboard, and later they paste it into another application.
In X windows the selection model is generalized to support more than one selection, and
they are identified by names like PRIMARY and CLIPBOARD. The CLIPBOARD selection is
used for copy and paste as in Macintosh and Windows. The PRIMARY selection is described
later. You could use other selection names, like SECONDARY or FOOBAR, but that only works
if the other applications know about that selection name. The selection data has both a type
and a format. These are described briefly later.
Data is not copied into a selection. Instead, an application asserts ownership of a selection,
and other applications request the value of the selection from that owner. This model is used
on all platforms. The window system keeps track of ownership, and applications are informed
when some other application takes away ownership. Several of the Tk widgets implement
selections and take care of asserting ownership and returning its value.
The X PRIMARY selection is used in a way that eliminates the explicit copy step in copy and
paste user actions. Whenever you select an object in your application, your application
automatically puts that value into the PRIMARY selection. The Tk entry, listbox, and text

widgets do this with their text selections, although you can turn this off with the
exportSelection widget attribute. Users typically insert the value of the PRIMARY
selection by clicking with the middle mouse button. There is only one instance of the
PRIMARY selection across all widgets and all applications. If the user makes a new selection
it automatically overwrites the previous value of the PRIMARY selection.

The CLIPBOARD is cross-platform.

If you want a mechanism that works on all platforms, use the CLIPBOARD selection. The
PRIMARY selection is implemented by Tk on all platforms, and you can use it within an
application, but on Windows and Macintosh the non-Tk applications do not know about the
PRIMARY selection. The main goal of copy and paste is to provide general interoperability
among all applications, so stick with the CLIPBOARD.
Tk 3.6 and earlier only supported the PRIMARY selection. When Tk 4.0 added support for the
CLIPBOARD, I tried to merge the two selections to "simplify" things for my users. Example
38-1 implements a Paste function that inserts either the PRIMARY or CLIPBOARD selection
into a text widget. The selection get command is used to retrieve the selection value:

Example 38-1. Paste the PRIMARY or CLIPBOARD selection

proc Paste { text } {
   if [catch {selection get} sel] {
      if [catch {selection get -selection CLIPBOARD} sel] {
         # no selection or clipboard data
         return
      }
   }
   $text insert insert $sel
}

This Paste function can be convenient, but it turns out that users still need to keep track of
the difference between the two selections. If a user only understands the CLIPBOARD, then
the use of PRIMARY is only surprising. I learned that it is best to have a separate paste user
action for the two selections. The convention is that <ButtonRelease-2> sets the insert
point and inserts the PRIMARY selection. (This convention is awkward with the one- and
two-button mice on Macintosh and Windows.) The <<Paste>> event (e.g., the Paste key)
simply inserts the CLIPBOARD selection at the current insert point. This convention is shown
in Example 38-2, although these bindings are defined automatically for the text and entry
widgets:

Example 38-2. Separate paste actions

bind Text <<Paste>> {
   catch {%W insert insert \
      [selection get -selection CLIPBOARD]
   }
}
bind Text <ButtonRelease-2> {
   %W mark set insert @%x,%y
   catch {%W insert insert \
      [selection get -selection PRIMARY]
   }
}

The selection Command
There are two Tcl commands that deal with selections. The selection command is a
general-purpose command that can set and get different selections. By default it manipulates
the PRIMARY selection. The clipboard command is a convenience command for
manipulating the CLIPBOARD selection.
The selection command exposes the fully general selection model of different selections,
types, and formats. You can define selection handlers that return selection values, and you
can assert ownership of a selection and find out when you lose ownership to another
application. Example 38-5 on page 596 shows a selection handler for a canvas.
A selection can have a type. The default is STRING. The type is different than the name of
the selection (e.g., PRIMARY or CLIPBOARD). Each type can have a format, and the default
format is STRING. Ordinarily these defaults are fine. If you are dealing with non-Tk
applications, however, you may need to ask for their selections by the right type (e.g.,
FILE_NAME). Formats include UTF8_STRING, STRING, ATOM, and INTEGER. An ATOM is a
name that is registered with the X server and identified by number. "Atoms and IDs" on page
667 describes Tk commands for manipulating atoms. It is probably not a good idea to use
non-STRING types and formats because it limits what other applications can use the
information. The details about X selection types and formats are specified in the Inter-Client
Communication Conventions Manual (David Rosenthal, Stuart Marks, X Consortium
Standard). This is distributed with the X11 sources and can be found on the web at http://
tronche.com/gui/x/icccm/.
All of the selection operations take a -selection option that specifies the name of the
selection being manipulated. This defaults to PRIMARY. Some of the operations take a -
displayof option that specifies what display the selection is on. The value for this option
is a Tk pathname of a window, and the selection on that window's display is manipulated.
This is useful in X where applications can have their windows on remote displays. The default
is to manipulate the selection on the display of the main window. Table 38-1 summarizes the
selection command:

Table 38-1. The selection command

selection clear ?-displayof win? ?-selection
sel?

Clears the specified selection.

selection get ?-displayof win? ?-selection
sel? ?-type type?

Returns the specified selection. The type defaults to STRING.

selection handle ?-selection sel? ?-type
type? ?-format format? window command

Defines command to be the handler for selection requests when
window owns the selection.

selection own ?-displayof window? ?-selection
sel?

Returns the Tk pathname of the window that owns the selection,
if it is in this application.

selection own ?-command command? ?-selection
sel? window

Asserts that window owns the sel selection. The command is
called when ownership of the selection is taken away from
window.

The clipboard Command
The clipboard command manipulates values in the CLIPBOARD selection. The
CLIPBOARD is meant for values that have been recently or temporarily deleted. It is use for
the copy and paste model of selections. Prior to Tk 8.4, you had to use the selection
command to retrieve values from the CLIPBOARD selection:

selection get -selection CLIPBOARD

However, Tk 8.4 introduced a clipboard get operation as a convenience for retrieving
the clipboard value.
Table 38-2 summarizes the clipboard command:

Table 38-2. The clipboard command

clipboard append ?-displayof win? ?-format
format? ?-type type? ?--? data

Appends data to the CLIPBOARD with the specified type and
format, which both default to STRING.

clipboard clear ?-displayof win? Clears the CLIPBOARD selection.

clipboard get ?-displayof win? ?-type type? Returns the CLIPBOARD selection. The type defaults to
STRING.

Selection Handlers

The selection handle command registers a Tcl command to handle selection requests.
The command is called to return the value of the selection to a requesting application. If the
selection value is large, the command might be called several times to return the selection
in pieces. The command gets two parameters that indicate the offset within the selection to
start returning data, and the maximum number of bytes to return. If the command returns
fewer than that many bytes, the selection request is assumed to be completed. Otherwise,

the command is called again to get the rest of the data, and the offset parameter is adjusted
accordingly.
You can also get a callback when you lose ownership of the selection. At that time it is
appropriate to unhighlight the selected object in your interface. The selection own
command sets ownership and registers a callback for when you lose ownership.

A Canvas Selection Handler
Example 38-3 through Example 38-7 implement cut and paste for a canvas. The
CanvasSelect_Demo procedure creates a canvas and sets up some bindings for cut and
paste:

Example 38-3. Bindings for canvas selection

proc CanvasSelect_Demo { c } {
   canvas $c
   pack $c
   $c create rect 10 10 50 50 -fill red -tag object
   $c create poly 100 100 100 30 140 50 -fill orange \
      -tag object
   # Set up cut and paste bindings
   $c bind object <Button-1> [list CanvasSelect $c %x %y]
   bind $c <Key-Delete> [list CanvasDelete $c]
   bind $c <<Cut>> [list CanvasCut $c]
   bind $c <<Copy>> [list CanvasCopy $c]
   bind $c <<Paste>> [list CanvasPaste $c]
   bind $c <Button-2> [list CanvasPaste $c %x %y]
   # Register the handler for selection requests
   selection handle $c [list CanvasSelectHandle $c]
}

The CanvasSelect procedure selects an object. It uses the find closest canvas
operation to find out what object is under the mouse, which works because the binding is
on canvas items with the object tag. If the binding were on the canvas as a whole, you
would use the find overlapping operation to limit selection to objects near the mouse
click. The CanvasHighlight procedure is used to highlight the selected object. It displays
small boxes at the corners of the object's bounding box. Finally, the CanvasSelectLose
procedure is registered to be called when another application asserts ownership of the
PRIMARY selection.

Chapter 38. Selections and the Clipboard> Page 5 Return to Table of Contents

Chapter 38. Selections and the Clipboard>
Practical Programming in Tcl and Tk, Fourth Edition By Ken Jones, Jeffrey Hobbs, Brent B. Welch ISBN:
0-13-038560-3 Publisher: Prentice Hall

Prepared for Ronald Fischer, Safari ID: ronald.fischer@fusshuhn.de

Print Publication Date: 2003/06/10 User number: 628024 Copyright 2007, Safari Books Online, LLC.
Reproduction, transmission and/or redistribution in any form by any means without the prior written permission from the publisher is prohibited.



Example 38-4. Selecting objects

proc CanvasSelect { w x y } {
   # Select an item on the canvas.
   global canvas
   set id [$w find closest $x $y]
   set canvas(select,$w) $id
   CanvasHighlight $w $id
   # Claim ownership of the PRIMARY selection
   selection own -command [list CanvasSelectLose $w] $w
   focus $w
}
proc CanvasHighlight {w id {clear clear}} {
   if {$clear == "clear"} {
      $w delete highlight
   }
   foreach {x1 y1 x2 y2} [$w bbox $id] { # lassign }
   foreach x [list $x1 $x2] {
      foreach y [list $y1 $y2] {
         $w create rectangle [expr $x-2] [expr $y-2] \
            [expr $x+2] [expr $y+2] -fill black \
            -tag highlight
      }
   }
}
proc CanvasSelectLose { w } {
   # Some other app has claimed the selection
   global canvas
   $w delete highlight
   unset canvas(select,$w)
}

Once you claim ownership, Tk calls back to the CanvasSelectHandle procedure when
another application, even yours, requests the selection. This uses CanvasDescription to
compute a description of the canvas object. It uses canvas operations to query the object's
configuration and store that as a command that will create the object:

Example 38-5. A canvas selection handler

proc CanvasSelectHandle { w offset maxbytes } {
   # Handle a selection request
   global canvas
   if ![info exists canvas(select,$w)] {
      error "No selected item"
   }
   set id $canvas(select,$w)
   # Return the requested chunk of data.
   return [string range [CanvasDescription $w $id] \
      $offset [expr $offset+$maxbytes]]
}
proc CanvasDescription { w id } {
   # Generate a description of the object that can
   # be used to recreate it later.
   set type [$w type $id]
   set coords [$w coords $id]
   set config {}
   # Bundle up non-default configuration settings
   foreach conf [$w itemconfigure $id] {
      # itemconfigure returns a list like
      # -fill {} {} {} red
      set default [lindex $conf 3]
      set value [lindex $conf 4]
      if {[string compare $default $value] != 0} {
         lappend config [lindex $conf 0] $value
      }
   }
   return [concat CanvasObject $type $coords $config]
}

The CanvasCopy procedure puts the description of the selected item onto the clipboard
with the clipboard append command. The CanvasDelete deletes an object and the
highlighting, and CanvasCut is built from CanvasCopy and CanvasDelete:

Example 38-6. The copy and cut operations

proc CanvasCopy { w } {
   global canvas
   if [info exists canvas(select,$w)] {
      set id $canvas(select,$w)
      clipboard clear
      clipboard append [CanvasDescription $w $id]
   }
}
proc CanvasDelete {w} {
   global canvas
   catch {
      $w delete highlight
      $w delete $canvas(select,$w)
      unset canvas(select,$w)
   }
}
proc CanvasCut { w } {
   CanvasCopy $w
   CanvasDelete $w
}

The CanvasPaste operation gets the value from the CLIPBOARD selection. The selection
value has all the parameters needed for a canvas create operation. It gets the position of

the new object from the <Button-2> event, or from the current mouse position if the
<<Paste>> event is generated. If the mouse is out of the window, then the object is just
put into the middle of the canvas. The original position and the new position are used to
compute values for a canvas move:

Example 38-7. Pasting onto the canvas

proc CanvasPaste { w {x {}} {y {}}} {
   # Paste the selection from the CLIPBOARD
   if [catch {selection get -selection CLIPBOARD} sel] {
      # no clipboard data
      return
   }
   if {[string length $x] == 0} {
      # <<Paste>>, get the current mouse coordinates
      set x [expr [winfo pointerx $w] - [winfo rootx $w]]
      set y [expr [winfo pointery $w] - [winfo rooty $w]]
      if {$x < 0 || $y < 0 ||
             $x > [winfo width $w] ||
             $y > [winfo height $w]} {
         # Mouse outside the window - center object
         set x [expr [winfo width $w]/2]
         set y [expr [winfo height $w]/2]
      }
   }
   if [regexp {^CanvasObject} $sel] {
      if [catch {eval {$w create} [lrange $sel 1 end]} id] {
         return;
      }
      # look at the first coordinate to see where to
      # move the object. Element 1 is the type, the
      # next two are the first coordinate
      set x1 [lindex $sel 2]
      set y1 [lindex $sel 3]
      $w move $id [expr $x-$x1] [expr $y-$y1]
   }
}

There is more you can do for a drawing program, of course. You'd like to be able to select
multiple objects, create new ones, and more. The ImPress application by Christopher Cox is
a full-featured page layout application based on the Tk canvas. You can find it on the Web
at:

http://www.ntlug.org/~ccox/impress/

