
Chapter 35. The Listbox Widget

The listbox provides a scrollable list of text lines. The listbox supports selections of one or
more lines.
Listbox widgets display a set of text lines in a scrollable display. The basic text unit is a line.
There are operations to insert, select, and delete lines, but there are no operations to modify
the characters in a line. As such, the listbox is suitable for displaying a set of choices, such as
in a file selection dialog. By default a user can select one item from a listbox, but you can
select multiple items by setting the selection mode attribute.

Using Listboxes

Manipulating Listbox Contents
The lines in a listbox are indexed from zero. The keyword index end addresses the last line.
Other indices are described on page 520. The most common programming task for a listbox
is to insert text. If your data is in a list, you can loop through the list and insert each element
at the end:

foreach item $list {
    $listbox insert end $item
}

You can insert several items at once. The next command uses eval to concatenate the list
onto a single insert command:

eval {$listbox insert end} $list

The delete operation deletes items from a listbox. You can delete either a single item or a
range of items:

$listbox delete 0     ;# Delete the first item
$listbox delete 3 5   ;# Delete items 3-5, inclusive

The listbox widget gained a listvariable option in Tk 8.3, which works analogously to
textvariable attributes in other widgets. You give listvariable the name of a
variable that contains a list value. Tk then keeps the value of the variable and the listbox

contents synchronized, so that each element in the variable's value is a line in the listbox.
Example 35-1 shows a simple example of modifying a listbox through its linked listvariable.

Example 35-1. Using -listvariable to link a listbox and variable

listbox .choices -height 5 -width 20 -listvariable states
pack .choices
lappend states Arizona
lappend states California "New Mexico"

Programming Listboxes
It is also common to react to mouse clicks on a listbox, although the default bindings handle
most of the details of selecting items. The nearest operation finds the listbox entry that is
closest to a mouse event. If the mouse is clicked beyond the last element, the index of the
last element is returned:

set index [$list nearest $y]

Example 35-2 displays two listboxes. The Scrolled_Listbox procedure on page 502 is
used to put scrollbars on the listboxes. When the user clicks on an item in the first listbox, it
is copied into the second listbox. When an item in the second listbox is selected, it is removed.
This example shows how to manipulate items selected from a listbox:

Example 35-2. Choosing items from a listbox

proc List_Select { parent values } {
   # Create two lists side by side
   frame $parent
   set choices [Scrolled_Listbox $parent.choices \
      -width 20 -height 5 ]
   set picked [Scrolled_Listbox $parent.picked \
      -width 20 -height 5]
   pack $parent.choices $parent.picked -side left \
      -expand true -fill both

   # Selecting in choices moves items into picked
   bind $choices <ButtonRelease-1> \
      [list ListTransferSel %W $picked]

   # Selecting in picked deletes items
   bind $picked <ButtonRelease-1> \
      {ListDeleteSel %W %y}

   # Insert all the choices
   foreach x $values {
      $choices insert end $x
   }
}
proc ListTransferSel {src dst} {
   foreach i [$src curselection] {
      $dst insert end [$src get $i]
   }
}
proc ListDeleteSel {w y} {
   foreach i [lsort -integer -decreasing [$w curselection]] {
      $w delete $i
   }
}
proc List_SelectValues {parent} {
   set picked $parent.picked.list
   set result {}
   foreach i [$w curselection] {
      lappend result [$w get $i]
   }
}
List_Select .f {apples oranges bananas \
             grapes mangos peaches pears}
pack .f -expand true -fill both

Bindings are created to move items from $choices to $picked, and to delete items from
$picked. Most of the work of selecting things in the listbox is done by the built-in bindings
on the Listbox binding tag. The different selection models are described on page 525.
Those bindings are on <ButtonPress-1> and <B1-Motion>. The selection is complete
by the time the <ButtonRelease-1> event occurs. Consider the
<ButtonRelease-1> binding for $choices:

bind $choices <ButtonRelease-1> \
    [list ListTransferSel %W $picked]

The list command is used to construct the Tcl command because we need to expand the
value of $picked at the time the binding is created. The command will be evaluated later
at the global scope, and picked will not be defined after the List_Select procedure
returns. Or, worse yet, an existing global variable named picked will be used, which is
unlikely to be correct!
Short procedures are used to implement the binding commands. This style has two
advantages. First, it confines the % substitutions done by bind to a single command. Second,
if there are any temporary variables, such as the loop counter i, they are hidden within the
scope of the procedure.
The ListTransferSel gets the list of all the selected items and loops over this list to insert
them into the other list. The ListDeleteSel procedure is similar. However, it sorts the
selection indices in reverse order. It deletes items from the bottom up so the indices remain
valid throughout the process.

The Listbox Widget
The listbox operations use indices to reference lines in the listbox. The lines are numbered
starting at zero. Keyword indices are also used for some special lines. The listbox keeps track
of an active element, which is displayed with underlined text. There is also a selection
anchor that is used when adjusting selections. Table 35-1 summarizes the keywords used for
indices.

Table 35-1. Listbox indices

0 Index of the first line.

active The index of the activated line.

anchor The index of the anchor point of the selection.

end Index of the last line.

number Index a line, counting from zero.

@x,y The line closest to the specified X and Y coordinates.

Table 35-2 presents the operations for programming a listbox. In the table, $w is a listbox
widget. Most of the operations have to do with the selection, and these operations are already
programmed by the default bindings for the Listbox widget class:

Table 35-2. Listbox operations

$w activate index Activates the specified line.

$w bbox index Returns the bounding box of the text in the
specified line in the form: xoff yoff
width height.

$w cget option Returns the value of the configuration option.

$w configure ... Queries or modifies the widget configuration.

$w curselection Returns a list of indices of the selected lines.

$w delete first?last? Deletes the lines from first to last, including the line at
last. The line at first is deleted if last is not given.

$w get first?last? Returns the lines from first to last as a list.

$w index index Returns the numerical index corresponding to index.

$w insert index ?string string string ...? Inserts the string items before the line at index. If index is
end, then append the items.

$w itemcget index option Returns the current value of the item's configuration option. (Tk
8.3)

$w itemconfigure index ?option? ?value? ?...? Queries or modifies the item's configuration options. (Tk 8.3)

$w nearest y Returns the index of the line closest to the widget-relative Y
coordinate.

$w scan mark x y Starts a scroll operation. x and y are widget-relative screen
coordinates.

$w scan dragto x y Scrolls from previous mark position.

$w see index Adjusts the display so the line at index is visible.

$w selection anchor index Anchors the selection at the specified line.

$w selection clear start?end? Clears the selection.

$w selection includes index Returns 1 if the line at index is in the selection.

$w selection set start ?end? Selects the lines from start to end.

$w size Returns the number of items in the listbox.

$w xview Returns the offset and span of visible contents. These are both
real numbers between 0 and 1.

$w xview index Shifts the display so the character at index is at the left edge of
the display.

$w xview moveto fraction Shifts the display so that fraction of the contents are off the
left edge of the display.

$w xview scroll num what Scrolls the contents horizontally by the specified number of
what, which can be units or pages.

$w yview Returns the offset and span of visible contents. These are both
real numbers between 0 and 1.

$w yview index Shifts the display so the line at index is at the top edge of the
display.

$w yview moveto fraction Shifts the display so that fraction of the contents are off the
top of the display.

$w yview scroll num what Scrolls the contents vertically by the specified number of what,
which can be units or pages.

As of Tk 8.3, the itemcget and itemconfigure operations allow you to control the
appearance of individual items in the listbox, overriding the listbox's defaults. Table 35-3 lists
the configuration options supported for individual items. Note that there is no direct support
for setting these values through the option resource database; you can set them only through
the listbox's itemconfigure operation.

Table 35-3. Listbox item configuration options

-background color Background color for the item.

-foreground color Foreground color for the item.

-selectbackground color Background color for the item when it's selected.

-selectforeground color Foreground color for the item when it's selected.

Listbox Bindings and Events
A listbox has an active element and it may have one or more selected elements. The active
element is highlighted according to the -activestyle options (by default, with an
underline), and the selected elements are highlighted with a different color. There are a large
number of key bindings for listboxes. You must set the input focus to the listbox for the key
bindings to work. Chapter 39 describes focus. There are four selection modes for a listbox,
and the bindings vary depending what mode the listbox is in. Table 35-4 lists the four possible
selectMode settings:

Table 35-4. The values for the selectMode of a listbox

single A single element can be selected.

browse A single element can be selected, and the selection can be dragged with the mouse. This is the default.

multiple More than one element can be selected by toggling the selection state of items, but you only select or deselect one
line at a time.

extended More than one element can be selected by dragging out a selection with the shift or control keys.

Browse Select Mode
In browse selection mode, <Button-1> selects the item under the mouse and dragging
with the mouse moves the selection, too. Table 35-5 gives the bindings for browse mode.

Table 35-5. Bindings for browse selection mode

<Button-1> Selects the item under the mouse. This becomes the active element, too.

<B1-Motion> Same as <Button-1>, the selection moves with the mouse.

<Shift-Button-1> Activates the item under the mouse. The selection is not changed.

<Key-Up> <Key-Down> Moves the active item up (down) one line, and selects it.

<Control-Home> Activates and select the first element of the listbox.

<Control-End> Activates and select the last element of the listbox.

<space> <Select> <Control-slash> Selects the active element.

Single Select Mode
In single selection mode, <Button-1> selects the item under the mouse, but dragging
the mouse does not change the selection. When you release the mouse, the item under that
point is activated. Table 35-6 specifies the bindings for single mode:

Table 35-6. Bindings for single selection mode

<ButtonPress-1> Selects the item under the mouse.

<ButtonRelease-1> Activates the item under the mouse.

<Shift-Button-1> Activates the item under the mouse. The selection is not changed.

<Key-Up> <Key-Down> Moves the active item up (down) one line. The selection is not changed.

<Control-Home> Activates and selects the first element of the listbox.

<Control-End> Activates and selects the last element of the listbox.

<space> <Select> <Control-slash> Selects the active element.

<Control-backslash> Clears the selection.

Extended Select Mode
In extended selection mode, multiple items are selected by dragging out a selection with
the first mouse button. Hold down the Shift key to adjust the ends of the selection. Use
the Control key to make a disjoint selection. The Control key works in a toggle fashion,
changing the selection state of the item under the mouse. If this starts a new part of the
selection, then dragging the mouse extends the new part of the selection. If the toggle action
cleared the selected item, then dragging the mouse continues to clear the selection. The
extended mode is quite intuitive once you try it. Table 35-7 specifies the complete set of
bindings for extended mode:

Table 35-7. Bindings for extended selection mode

<Button-1> Selects the item under the mouse. This becomes the anchor point for adjusting the selection.

<B1-Motion> Sweeps out a selection from the anchor point.

<ButtonRelease-1> Activates the item under the mouse.

<Shift-Button-1> Adjusts the selection from the anchor item to the item under the mouse.

<Shift-B1-Motion> Continues to adjust the selection from the anchor.

<Control-Button-1> Toggles the selection state of the item under the mouse, and makes this the anchor point.

<Control-B1-Motion> Sets the selection state of the items from the anchor point to the item under the mouse to be
the same as the selection state of the anchor point.

<Key-Up> <Key-Down> Moves the active item up (down) one line, and starts a new selection with this item as the anchor
point.

<Shift-Up> <Shift-Down> Moves the active element up (down) and extends the selection to include this element.

<Control-Home> Activates and selects the first element of the listbox.

<Control-Shift-Home> Extends the selection to the first element.

<Control-End> Activates and selects the last element of the listbox.

<Control-Shift-End> Extends the selection to the last element.

<space> <Select> Selects the active element.

<Escape> Cancels the previous selection action.

<Control-slash> Selects everything in the listbox.

<Control-backslash> Clears the selection.

Multiple Select Mode
In multiple selection mode you can select more than one item, but you can add or remove
only one item at a time. Dragging the mouse does not sweep out a selection. If you click on
a selected item it is deselected. Table 35-8 specifies the complete set of bindings for
multiple selection mode.

Table 35-8. Bindings for multiple selection mode

<Button-1> Selects the item under the mouse.

<ButtonRelease-1> Activates the item under the mouse.

<Key-Up> <Key-Down> Moves the active item up (down) one line, and starts a new selection with this item as the anchor
point.

<Shift-Up> <Shift-Down> Moves the active element up (down).

<Control-Home> Activates and selects the first element of the listbox.

<Control-Shift-Home> Activates the first element of the listbox.

<Control-End> Activates and selects the last element of the listbox.

<Control-Shift-End> Activates the last element of the listbox.

<space> <Select> Selects the active element.

<Control-slash> Selects everything in the listbox.

<Control-backslash> Clears the selection.

Scroll Bindings
There are several bindings that scroll the display of the listbox. In addition to the standard
middle-drag scrolling, there are some additional key bindings for scrolling. Table 35-9
summarizes the scroll-related bindings:

Table 35-9. Listbox scroll bindings

<Button-2> Marks the start of a scroll operation.

<B2-Motion> Scrolls vertically and horizontally.

<MouseWheel> Scrolls vertically.

<Button-4> Mousewheel support on Unix only; scrolls up. (Tk 8.3)

<Button-5> Mousewheel support on Unix only; scrolls down. (Tk 8.3)

<Left> <Right> Scrolls horizontally by one character.

<Control-Left> <Control-Right>
<Control-Prior> <Control-Next>

Scrolls horizontally by one screen width.

<Prior> <Next> Scrolls vertically by one screen height.

<Home> <End> Scrolls to left and right edges of the screen, respectively.

Listbox Virtual Events
As of Tk 8.1, the listbox widget generates a <<ListboxSelect>> virtual event whenever
the listbox selection changes. The event fires after the selection has changed, so the binding
action can access the new selection. The easiest way to be aware of changes to the listbox
selection is to bind to this virtual event, as shown in Example 35-3:

Example 35-3. Using the <<ListboxSelect>> virtual event

proc ListboxChanged {w} {
    puts -nonewline "Listbox $w selection is now: "
    foreach index [$w curselection] {
        puts -nonewline "[$w get $index] "
    }
    puts ""
}
bind .lbox <<ListboxSelect>> {ListboxChanged %W}

Listbox Attributes
Table 35-10 lists the listbox widget attributes. The table uses the resource name for the
attribute, which has capitals at internal word boundaries. In Tcl commands these options are
specified with a dash and all lowercase.

Table 35-10. Listbox attribute resource names

activeStyle Display style of the active element: dotbox, underline (default), or
none. (Tk 8.4)

background Background color (also bg).

borderWidth Extra space around the edge of the text.

cursor Cursor to display when mouse is over the widget.

disabledForeground Foreground color when the widget is disabled. (Tk 8.4)

exportSelection If true, then the selected text is exported via the X selection mechanism.

font Font for the text.

foreground Foreground color (also fg).

height Number of lines in the listbox.

highlightBackground Focus highlight color when widget does not have focus.

highlightColor Focus highlight color when widget has focus.

highlightThickness Thickness of focus highlight rectangle.

listVariable Name of Tcl variable containing a list whose value is synchronized with the widget, where each list
element is a line of the listbox. (Tk 8.3)

relief flat, sunken, raised, groove, solid, or ridge.

selectBackground Background color of selection.

selectForeground Foreground color of selection.

selectBorderWidth Width of selection border. Nonzero for 3D effect.

selectMode Mode: browse, single, extended, or multiple.
setGrid Boolean. Set gridding attribute.

state Widget state: normal or disabled. (Tk 8.4)

takeFocus Controls focus changes from keyboard traversal.

width Width, in average character sizes.

xScrollCommand Connects listbox to a horizontal scrollbar.

yScrollCommand Connects listbox to a vertical scrollbar.

Geometry Gridding
The setGrid attribute affects interactive resizing of the window containing the listbox. By
default, a window can be resized to any size. If gridding is turned on, the size is restricted so
that a whole number of lines and a whole number of average-width characters is displayed.
Gridding affects the user feedback during an interactive resize. Without gridding the size is
reported in pixel dimensions. When gridding is turned on, then the size is reported in gridded
units.

